
Tropical computations for toric intersection theory in Macaulay2

ALESSIO BORZÌ

ABSTRACT: We present the Macaulay2 package TropicalToric for toric intersection theory computa-
tions using tropical geometry.

1. INTRODUCTION. Toric varieties are ubiquitous in algebraic geometry. Their intersection theory
was first studied in Fulton and Sturmfels [12], and it has many applications in different contexts, in-
cluding wonderful and tropical compactifications [7; 10; 34], birational geometry [5; 14; 15], tropical
geometry [23], tropical intersection theory [2; 21; 22; 29; 30] and combinatorial Hodge theory [1; 19; 20].

In a certain way, intersection classes of a toric variety with fan 6 can be thought of in terms of balanced
subfans of 6, also referred to as Minkowski weights; see [12] or [27, Theorem 6.7.5]. From the structure
theorem of tropical geometry [27, Theorem 3.3.5], we know that the tropicalization of a subvariety of a
torus Y ⊆ T n is a balanced fan. A surprising connection between tropical and toric geometry is that the
tropicalization of Y is the balanced fan corresponding to the intersection class of the closure of Y inside
an “enough refined” toric variety; see Theorem 3.1 for a more precise statement. This fact allows us to
compute toric intersection classes starting from the data of the tropicalization.

We present a new package, TropicalToric, for Macaulay2 [26]. The package implements toric
cycles and intersection products on simplicial toric varieties (Section 2), and, following the ideas outlined
above, allows us to compute the intersection class of an irreducible subvariety of a simplicial toric variety
not contained in the toric boundary, from the data of its tropicalization (Section 3). The tropicalization is
performed with the use of the Macaulay2 package Tropical [3]. Further, we present some applications
to the intersection theory of wonderful compactifications and the moduli space M0,n and illustrate an
example in a multiprojective space using a theorem of Huh and Katz [20] about characteristic polynomials
of realizable matroids.

2. TORIC INTERSECTION THEORY. In this section, we review the basics of toric intersection theory,
for more information see [12], [6, Section 12.5] or [27, Section 6.7]. In addition, we showcase how it is
implemented in the package.

Let X6 be a smooth complete toric variety of dimension n with fan 6. We denote by 6(k) the
cones of 6 of dimension k, with Z k(X6) = Zn−k(X6) the group of codimension k cycles and with
Ak(X6) = An−k(X6) the codimension k Chow group, that is, the group of codimension k cycles modulo

rational equivalence. The codimension k Chow group Ak(X6) is generated by the set {[V (σ )] : σ ∈ 6(k)}

of classes of orbit closures of codimension k, and the relations in each Chow group can be described in
an explicit way; see [12, Proposition 2.1].

We can show (see [11, Chapter 8]) that there is an intersection product Ak(X6)× Ar (X6)→ Ak+r (X6)

that makes A∗(X6) =
⊕n

k=0 Ak(X6) into a graded ring, called the Chow ring of X6 . If we now assume
that X6 is just complete and simplicial, then the intersection product can be defined on rational cycles,
making A∗(X6)Q = A∗(X6) ⊗ Q into a graded ring. The structure of the Chow ring has an explicit
description; see, for instance, [27, Theorem 6.7.1].

Our Macaulay2 package implements toric cycles and the intersection product as in [6, Lemma 12.5.2].

Example 2.1. Let X6 be the blow-up of P2 at one of the coordinate points, where the fan 6 and the
primitive ray vectors of its rays are

ρ0

ρ1ρ2

ρ3

ρ0 ρ1 ρ2 ρ3[
1 1 0 −1
0 1 1 −1

]

Let H be the strict transform in X6 of a general line in P2 and E be the exceptional divisor. The
Picard group of X6 is generated by the classes of these two divisors Pic(X6) = ⟨[H ], [E]⟩. With the
notation above, we have

[V (ρ0)] = [H ] − [E], [V (ρ1)] = [E], [V (ρ2)] = [H ] − [E], [V (ρ3)] = [H ].

Now we verify with our package that the divisor class [V (ρ1)] has a negative self-intersection.
i1 : needsPackage "TropicalToric";

i2 : raysList = {{1,0},{1,1},{0,1},{-1,-1}};

i3 : coneList = {{0,1},{1,2},{2,3},{3,0}};

i4 : X = normalToricVariety (raysList, coneList);

Now define the toric cycle V (ρ1).
i5 : E = X_{1}

o5 = X
{1}

o5 : ToricCycle on X

We point out that the type ToricCycle should not be confused with the type ToricDivisor from the
NormalToricVarieties package. The toric cycle V (σ ) of the normal toric variety X associated to the
cone σ given by a list of rays L is defined with the command X_L. For example, X_{1,2} or X_{0} define
toric cycles, whereas X_1 defines a toric divisor. We are allowed only to multiply a toric cycle with a
toric divisor. Now, we finally compute the self intersection of E :

i6 : X_1 * E

o6 = - X
{1, 2}

o6 : ToricCycle on X

The resulting cycle −V (ρ1 + ρ2) is rationally equivalent to E2. The negative sign tells us that the self-
intersection number of the exceptional divisor is −1. We can compute the degree of maximal codimension
cycles with degCycle:

i7 : degCycle(-X_{1,2})

o7 = -1

3. TROPICAL COMPUTATIONS. In this section, we describe and showcase the algorithm implemented
in the main function of the package classFromTropical that computes the intersection class of an
irreducible subvariety of a smooth toric variety from its tropicalization.

The algorithm is mainly based on the following result, that appears in various versions in the literature;
see, for instance, [23, Lemma 2.3], [21, Section 9] or [27, Theorem 6.7.7].

Theorem 3.1. Let Y be a subvariety of the algebraic torus T n , and let Y be its closure in a toric variety
X6 such that |6| = trop(Y ) and 6 is simplicial. Let 6′ be a simplicial completion of the fan 6, and let
i : X6 → X6′ be the induced inclusion. Then, for every maximal cone σ in 6, we have

m(σ ) = deg
(
[i∗(Y )] · [V (σ )]

)
,

where m(σ ) is the multiplicity of σ in trop(Y ).

Now let Y be an irreducible k-dimensional subvariety of an n-dimensional simplicial toric variety X6 ,
and suppose that Y ∩ T n

̸= ∅. Note that in this setting we cannot directly apply Theorem 3.1 since
trop(Y ) is not necessarily a subfan of 6.

In order to compute the class of Y in the Chow ring of X6 , we proceed as follows. First, let 6′

be a completion of 6 and i : X6 → X6′ be the induced inclusion. Now let 6̃ be a refinement of 6′

such that it contains a subfan with support the tropicalization of Y ∩ T n , and let π : X6̃ → X6′ be the
induced toric map. From [12, Proposition 2.4], we have an isomorphism Ak(X6′) ≃ Hom(Ak(X6′), Z)

mapping a class [Z ] to the homomorphism [Z ′
] 7→ deg([Z ] · [Z ′

]). Therefore, in order to compute the
class [Y ] ∈ Ak(X6), it is enough to compute the intersection numbers deg([i∗(Y )] · [V (σ )]) for every
σ ∈ 6′(k), as the classes [V (σ )] generate Ak(X6′). Let Y ′ be the strict transform of Y in X6̃ . From the
projection formula [11, Proposition 2.3 (c)], we have

[i∗(Y )] · [V (σ )] = π∗([Y ′
]) · [V (σ )] = π∗

(
[Y ′

] ·π∗([V (σ )])
)
,

from which it follows that deg
(
[i∗(Y )]·[V (σ )]

)
= deg

(
[Y ′

]·π∗([V (σ )])
)
. These last intersection numbers

can be computed from the tropicalization of Y ∩ T n by using Theorem 3.1, since deg
(
[Y ′

] · [V (σ ′)]
)

is
the multiplicity of the cone σ ′

∈ 6̃(k) in the tropicalization of Y ∩ T n .
The algorithm described above, while working on any simplicial toric variety, requires to compute a

completion. This can be avoided by requiring the toric variety X6 to be smooth. In fact, the only step
in which we are really using the completion is when we apply [12, Proposition 2.4] (sometimes called
Kronecker duality). If the variety X6 is smooth, this can be substituted instead with Poincaré duality.

The function classFromTropical performs the above algorithm to compute a toric cycle rationally
equivalent to a given irreducible subvariety Y of a smooth toric variety X6 (by using Poincaré duality).
The input of the function consists of the toric variety X6 and the ideal I of Y ∩ T n of the Laurent
ring of T n . Since Laurent rings are not implemented in Macaulay2, the actual input will be instead the
saturation of I with respect to the product of the variables in the polynomial ring:

i2 : X = toricProjectiveSpace 2;

i3 : R = QQ[x,y];

i4 : I = ideal(x+y+1);

i5 : classFromTropical(X,I)

o5 = X
{0}

o5 : ToricCycle on X

i6 : J = ideal(x*y + x + y);

i7 : classFromTropical(X,J)

o7 = 2*X
{0}

o7 : ToricCycle on X

The function classFromTropicalCox allows us to input the ideal of Y in the Cox ring of X6:
i8 : R = ring X;

i9 : I = ideal(R_0+R_1+R_2);

i10 : classFromTropicalCox(X,I)

o10 = X
{0}

o10 : ToricCycle on X

4. APPLICATIONS.

4A. Wonderful compactifications. Let A be an essential hyperplane arrangement of n + 1 hyperplanes
in Pd . The intersection lattice L(A) of A is isomorphic to the lattice of flats of the underlying matroid M
of A [32, Proposition 3.6]. Fix a building set G of the lattice of flats of M (see [10, Section 2]), let
6 ⊆ Rn+1/R1 ≃ Rn be the Bergman fan of M with respect to G (see [27, Chapter 4]) and let X6

be its associated toric variety. From [27, Proposition 4.1.1], the hyperplane arrangement complement
Y = Pd

\∪A is naturally isomorphic to a linear subspace of the algebraic torus T n . Thus we can embed Y
inside the toric variety X6 and consider its closure Y . This compactification coincides with the so-called
De Concini–Procesi wonderful compactification [7], with respect to the building set G (see [34, Section 4]).
The next result follows from [7, Theorem 3.2]; see also [9, Definition 2.3].

Proposition 4.1. Let X1, . . . , X t be a linear extension of the opposite order of L(A). The wonderful
compactification Y is the result of successively blowing up Pd at (the strict transforms of ) X1, . . . , X t .

In [10], Feichtner and Yuzvinsky showed that the cohomology of Y agrees with that of X6 . Since both
varieties are homology isomorphism schemes (in the sense of the definition in the appendix of [24]), their
Chow rings coincide as well.

Theorem 4.2 [27, Theorem 6.7.14]. Let Y be a wonderful compactification of a hyperplane arrangement
A with respect to a building set G, and let 6 be the associated Bergman fan. Then

A∗(Y ) ≃ A∗(X6),

where the above isomorphism is the pullback map induced by the inclusion.

The above theorem allows us to view the intersection classes of a wonderful compactification as
intersection classes of the associated toric variety. Thus, we can use our package to perform intersection
theory computations on wonderful compactifications.

Example 4.3. Let A be a line arrangement consisting of 4 lines L0, L1, L2, L3 in P2 given by the
equations x0 = 0, x1 = 0, x2 = 0, x0 + x1 = 0, respectively. Let A be the matrix with columns the normal
vectors of the lines L i , and let P1, P2, P3, P4 be the points of intersection of the lines of A, depicted as

A =

1 0 0 1
0 1 0 1
0 0 1 0


L0

L3

L1

L2
P3

P4

P2

P1

The underlying matroid M of A, on the ground set {0, 1, 2, 3}, is realized by the matrix A by labeling the
columns with 0, 1, 2, 3, respectively. The lattice of flats L(M) of M is represented by

∅

{0} {1} {2} {3}

{0, 1, 3} {0, 2} {1, 2} {2, 3}

{0, 1, 2, 3}

There are four rank 1 flats, corresponding to the lines L0, L1, L2, L3, and four rank 2 flats, corresponding
to the points P1, P2, P3, P4. Let G = L(M) \ {∅} be the maximal building set of L(M). Then, from
Proposition 4.1, the wonderful compactification Y of the complement Y = P2

\ ∪A with respect to G is
the blow-up of P2 at the points P1, P2, P3, P4. In particular, Y is a smooth projective surface, all Weil
divisors are Cartier [17, Proposition II.6.11] and the class group is isomorphic to the Picard group [17,
Corollary II.6.16]. From [17, Proposition V.3.2], the Picard group of Y has a basis given by

Pic(Y ) = ⟨[H ], [E1], . . . , [Et ]⟩, (1)

where [H ] is the class of the strict transform H of a general line in P2, and [Ei ] is the class of the
exceptional divisor Ei of the blow-up at Pi .

The Bergman fan 6 ⊆ R4/R1 of M with respect to G has eight rays, denoted {ρi : 0 ≤ i ≤ 7}. Their
primitive ray vectors are given by the columns of the matrix

ρ0 ρ1 ρ2 ρ3 ρ4 ρ5 ρ6 ρ7
1 0 0 0 1 1 0 0
0 1 0 0 1 0 1 0
0 0 1 0 0 1 1 1
0 0 0 1 1 0 0 1


where ρ0, ρ1, ρ2, ρ3 correspond to the rank 1 flats in G, which in turn correspond to the lines L0, L1,
L2, L3, respectively, and ρ4, ρ5, ρ6, ρ7 correspond to the rank 2 flats in G, that correspond to the points
P1, P2, P3, P4, respectively. Since G is the maximal building set, the maximal cones of 6 are just the
maximal chains of the lattice of flats L(M).

By using the isomorphism in Theorem 4.2, let [Yρi ] denote the class in A∗(Y ) isomorphic to the class
of the torus invariant divisor of X6 associated to the ray ρi . Expressing these divisors in the Picard
basis (1), we have

[Yρ0] = [H ] − [E1] − [E2], [Yρ4]= [E1],

[Yρ1] = [H ] − [E1] − [E3], [Yρ5]= [E2],

[Yρ2] = [H ] − [E2] − [E3] − [E4], [Yρ6]= [E3],

[Yρ3] = [H ] − [E1] − [E4], [Yρ7]= [E4].

(2)

Let C[y±1
0 , y±1

1 , y±1
2 ] be the Laurent ring of the torus

T 3
= {(1 : y0 : y1 : y2) : y0, y1, y2 ∈ C∗

} ⊆ P3.

The embedding Y ↪→ T 3 is given by (x0 : x1 : x2) 7→ (x0 : x1 : x2 : x0 + x1), and the Laurent ideal of Y
inside T 3 is I = (−1 − y0 + y2).

Now, let C be the conic in P2 passing through the points P1, P2 and P3 given by the equation
x0x1 + x0x2 + x1x2. The ideal of C in T 3 is (y0 + y1 + y0 y1) + I . We expect the class of its strict
transform in Y to be [2H − E1 − E2 − E3]. We now verify this with our package, using the function
classWonderfulCompactification.

i2 : R = QQ[y_0,y_1,y_2];

i3 : I = ideal(-1-y_0+y_2);

i4 : f = y_0+y_1+y_0*y_1;

i5 : raysList = {{-1,-1,-1},{1,0,0},{0,1,0},
{0,0,1},{0,-1,0},{-1,0,-1},
{1,1,0},{0,1,1}};

i6 : conesList = {{4,0},{4,1},{4,3},{5,0},{5,2},
{6,1},{6,2},{7,2},{7,3}};

i7 : X = normalToricVariety (raysList, conesList);

i8 : D = classWonderfulCompactification(X,I,f)

o8 = X + X + X
{0} {4} {1}

o8 : ToricCycle on X

To check that this is the result we expect, compare with (2). Note that we have (tropically) dehomogenized
the rays of X6 with respect to the first coordinate in order to be consistent with our choice of coordinates
of T 3.

4B. The moduli space M0,n. The Deligne–Mumford compactification of the moduli space M0,n can
be realized as a wonderful compactification; see, for instance, [27, Example 6.7.16]. Therefore, we can
apply to M0,n the machinery described in the previous section. As an application, we compute one of the
15 Keel–Vermeire divisors of M0,6, using one of the equations listed in [16, Table 2]. These divisors,
found independently by Keel and Vermeire [35], were the first example of an effective divisor of M0,n

whose class lies outside the cone generated by the classes of the boundary divisors, answering in the
negative a conjecture of Fulton; see [25].

i2 : R = QQ[x_0..x_8];

i3 : I = ideal {-x_0+x_3+x_4, -x_1+x_3+x_5,-x_2+x_3+x_6,
-x_0+x_2+x_7, -x_1+x_2+x_8, -x_0+x_1+1};

i4 : X = normalToricVariety fan tropicalVariety I;

i5 : f = x_0*x_1-x_2*x_3;

i6 : D = classWonderfulCompactification(X,I,f);

i7 : D = toricDivisorFromCycle(D)

o7 = X - X - 2*X + X + 2*X + 2*X - X + 2*X + 2*X - X
2 5 6 7 9 10 11 13 14 17

o7 : ToricDivisor on X

Now fix the Picard basis of X6 given by the boundary divisors associated to the rays of 6 with primitive
ray vectors not equal to the standard vectors ei . The complement of this Picard basis is indexed by the list
l = {0, 1, 2, 4, 5, 7, 11, 13, 21}. The function makeTransverse computes a divisor linearly equivalent to
a given divisor D, with support disjoint from a given list l. We use this function to compute a representation
of the class of the Keel–Vermeire divisor computed above, in the Picard basis we fixed:

i8 : l = {0,1,2,4,5,7,11,13,21};

i9 : D = makeTransverse(D,l)

o9 = X - X - X + X + 2*X - X - X + X + 2*X + 2*X
3 6 8 9 10 14 16 17 18 19
+ 2*X - X - X

20 22 24

o9 : ToricDivisor on X

Finally, we verify that the obtained divisor is outside the cone generated by the classes of boundary divisors.
In order to do so, we interface with Polymake [13] by using the function polymakeConeContains:

i10 : D = apply(#rays X, i->D#i);

i11 : Bdivisors = apply(#rays X, i-> makeTransverse(X_i,l));

i12 : Bdivisors = apply(Bdivisors, B-> apply(#rays X, i->B#i));

i13 : polymakeConeContains(D,Bdivisors)

o13 = false

In [18] it was proved, by using computational methods, that the boundary divisors and the Keel–
Vermeire divisors generate the effective cone of M0,6. In [5] it was proved that the effective cone of M0,n

for n ≥ 10 is not polyhedral. The problem of determining the effective cone of M0,n for n ∈ {7, 8, 9}

is still open. Some examples of extremal effective divisors on M0,7 were found in [4; 8; 28]. We
performed computations similar to those displayed above on M0,7 and found the mentioned examples
with a brute-force approach. More recently, in [31], several thousands of extremal effective divisors
on M0,7 were found.

4C. Characteristic polynomials. Our last application is an explicit verification of a theorem proved by
Huh and Katz [20] about characteristic polynomials of realizable matroids.

Let A be an arrangement of n + 1 hyperplanes on Pd , let M be its underlying matroid of rank d + 1,
and let L(M) be the lattice of flats of M . The characteristic polynomial of M is

χM(q) =

∑
F∈L(M)

µ(∅, F)qd+1−r(F),

where µ is the Möbius function of L(M); see [33, Section 3.7]. The reduced characteristic polynomial
of M is χ M(q) = χM(q)/(q − 1).

Now we embed the complement Y = Pd
\∪A in T n

⊆ Pn , as described in Section 4A, and consider
the Cremona map

ϕ : Pn 99K Pn, (x0, . . . , xn) 7→ (x−1
0 , . . . , x−1

n ).

Finally, let Z be the closure in Pn
× Pn of the graph Z of the restriction ϕ|Y .

Theorem 4.4 (Huh and Katz [20]). Define the integers ai ∈ Z by the formula

χ M(q) =

d∑
i=0

(−1)i ai qd−i .

Then

[Z ] =

d∑
i=0

ai [P
r−i

× Pi
] ∈ Ad(Pn

× Pn).

Example 4.5. Let G be the graph

G = A =

1 0 0 1 1
0 1 0 −1 0
0 0 1 0 −1

 .

Let M be the rank 3 graphic matroid of G, realized by the matrix A above. The characteristic polynomial of
M coincide with the chromatic polynomial of G. Let C[x±1

0 , x±1
1 , x±1

2 , x±1
3 ] be the Laurent ring of the torus

T 4
= {(1 : x0 : x1 : x2 : x3) : x0, x1, x2, x3 ∈ C∗

} ⊆ P4.

Let A be the hyperplane arrangement realizing M . More explicitly, the normal vectors of its hyperplanes
are the columns of the matrix A. The Laurent ideal of the hyperplane arrangement complement Y =P2

\∪A
embedded in T 4

⊆ P4 is given by I = (−1 + x0 + x2, −1 + x1 + x3).

Now consider a copy of T 4 with Laurent ring C[x±1
4 , x±1

5 , x±1
6 , x±1

7 ]. Let Z ⊆ T 4
× T 4 be the graph

of the Cremona map ϕ : P4 99K P4 restricted to Y . The ideal of Z in the Laurent ring C[x±1
0 , . . . , x±1

7 ]

of T 4
× T 4 is generated by I and the polynomials xi xi+4 − 1 for i ∈ {0, 1, 2, 3}.

i2 : R = QQ[x_0..x_7];

i3 : I = ideal(-1+x_0+x_2,-1+x_1+x_3,
x_0*x_4-1,x_1*x_5-1,x_2*x_6-1,x_3*x_7-1);

o3 : Ideal of R

i4 : P4 = toricProjectiveSpace 4;

i5 : X = NormalToricVarieties$cartesianProduct(P4,P4);

i6 : D = classFromTropical(X,I)

o6 = 4*X + 4*X + X
{0,1,2,3,5,6} {0,1,2,5,6,7} {0,1,5,6,7,8}

o6 : ToricCycle on X

We obtained [Z ] = [P2
× P0

] + 4[P1
× P1

] + 4[P0
× P2

]. We now verify that the coefficients of this
class are the same, up to sign, to those of the (reduced) chromatic polynomial of G:

i7 : needsPackage "Graphs";

i8 : G = graph({{0,1},{1,2},{2,3},{3,0},{0,2}});

i9 : p = chromaticPolynomial G
4 3 2

o9 = x - 5x + 8x - 4x

o9 : ZZ[x]

i10 : x = (ring p)_0;

i11 : p/(x-1)
3 2

o11 = x - 4x + 4x

o11 : frac(ZZ[x])

